

Quantifier Elimination with *Mathematica* #2

ー東大入試を 2行で 解くー

2018 年7月 Cabri研究会 生越 茂樹

QEのコマンド

$f \vee g$	$f \parallel g$ (or)
$f \wedge g$	$f \&\& g$ (and)
$\neg f$	$! f$ (not)
$f \rightarrow g$	$\text{Implies}[f,g]$ ($\Rightarrow$)
$f \Leftrightarrow g$	$\text{Equivalent}[f,g]$ (equiv)
$\exists x, f$	$\text{Exists}[x,f]$
$\forall x, f$	$\text{ForAll}[x,f]$
$x \in A$	$\text{Element}[x,A]$ (elem)
p のQE	$\text{Resolve}[p,\text{Reals}]$ または $\text{Reduce}[p,\text{Reals}]$

- 通常は Resolve の代わりに Reduce でも大丈夫(内部で Resolve をcall). しかし時間が掛かる.
- $\vee, \wedge, \Rightarrow$ などは“Esc”を使って挟むと入力できる. 例えば “Esc”と”=”と”>”と”Esc” で $\Rightarrow$ となる.
- QEはCADを応用している. CAD(Cylindrical Algebraic Decomposition)とは, 空間を符号が一定な部分集合に分割すること. なお CADはQE以外にも有効で *Mathematica* のコマンドは「 $\text{GenericCylindricalDecomposition}$ 」と「 $\text{CylindricalDecomposition}$ 」. (後者が数学的な厳密解)
- CADの順序の変更が可能. 例えば「 $\text{Resolve}[p,\{y,x\},\text{Reals}]$ 」とすると $x \square g(y)$ の形になる. 「 $\text{Reduce}[p,\{y,x\},\text{Reals}]$ 」より速いことが多い.
- 結果の図示は RegionPlot や RegionPlot3D を使う. 測度は RegionMeasure が大体使える.

§1. いろいろな方程式

[異なる2実数解を持つ条件]

a, b, c, d を正の数とする. 不等式 $\begin{cases} s(1-a)-tb > 0 \\ -sc+t(1-d) > 0 \end{cases}$ を同時にみたす

正の数 s, t があるとき, 2次方程式 $x^2 - (a+d)x + (ad-bc) = 0$ は $-1 < x < 1$ の範囲に異なる2つの実数解を持つことを示せ (1996年文理)

```
g[x_] = x^2 - (a + d) x + (a d - b c);
```

```
ForAll[{a, b, c, d, s, t}, a > 0 && b > 0 && c > 0 && d > 0 && s > 0 && t > 0 && s (1 - a) - t b > 0 && -s c + t (1 - d) > 0] =>
```

```
Exists[{x1, x2}, g[x1] == g[x2] == 0 && -1 < x1 < x2 < 1];
```

```
Resolve[%, Reals]
```

```
True
```

[複素数解を持ちうる場合]

定数 p に対して, 3次方程式 $x^3 - 3x - p = 0$ の実数解の中で最大のものと最小のものの積を $f(p)$ とする. ただし, 実数解がただひとつのときには, その2乗を $f(p)$ とする.

(1) p がすべての実数を動くとき, $f(p)$ の最小値を求めよ.

(2) p の関数 $f(p)$ のグラフの概形をえがけ. (1991年理)

(* (1) はExists{x1,x2,x3,p}とする*)

Clear[x1, x2, x3, M, p]

Exists[{x1, x2, x3, p}, (x1 | x2 | x3 | p | M) ∈ Reals,

$x1 + x2 + x3 == 0 \&\& x1 x2 + x2 x3 + x3 x1 == -3 \&\& x1 x2 x3 == p \&\& x1 \leq x2 \leq x3 \&\& x1 + x3 == M] \vee$

Exists[{x1, x2, x3, p}, (x1 | p | M) ∈ Reals && ! x2 ∈ Reals && ! x3 ∈ Reals &&

$x1 + x2 + x3 == 0 \&\& x1 x2 + x2 x3 + x3 x1 == -3 \&\& x1 x2 x3 == p \&\& x1 + x1 == M];$

Resolve[

%,

Complexes]

$-3 \leq M \leq -2 \mid \mid M > 4$

(* (2) はExists{x1,x2,x3}とする*)

Exists[{x1, x2, x3}, (x1 | x2 | x3 | p | M) ∈ Reals, $x1 + x2 + x3 == 0 \&\& x1 x2 + x2 x3 + x3 x1 == -3 \&\& x1 x2 x3 == p \&\& x1 \leq x2 \leq x3 \&\& x1 + x3 == M] \vee$

Exists[{x1, x2, x3},

$(x1 | p | M) \in \text{Reals} \&\& ! x2 \in \text{Reals} \&\& ! x3 \in \text{Reals} \&\& x1 + x2 + x3 == 0 \&\& x1 x2 + x2 x3 + x3 x1 == -3 \&\& x1 x2 x3 == p \&\& x1 + x1 == M];$


Resolve[%, Complexes]

RegionPlot[ImplicitRegion[%, {{p, -5, 5}, {M, -3, 7}}], Axes → True]

$(p \in \text{Reals} \&\& ((M < 0 \&\& p \geq 0 \&\& 3 M^2 + M^3 - p^2 == 0 \&\& M^3 + 2 p^2 == 0) \mid \mid (M < 0 \&\& p \leq 0 \&\& 3 M^2 + M^3 - p^2 == 0 \&\& M^3 + 2 p^2 == 0) \mid \mid$

$(M < 0 \&\& p > 0 \&\& 3 M^2 + M^3 - p^2 == 0 \&\& M^3 + 2 p^2 == 0) \mid \mid (M < 0 \&\& 3 M^2 + M^3 - p^2 == 0 \&\& M^3 + 2 p^2 \leq 0) \mid \mid$

$(M < 0 \&\& p > 0 \&\& 3 M^2 + M^3 - p^2 == 0 \&\& M^3 + 2 p^2 \leq 0)) \mid \mid (p \in \text{Reals} \&\& M > 4 \&\& 9 M - 6 M^2 + M^3 - p^2 == 0)$




§2. 最大・最小値

[正方形が $y \geq x^2$ に含まれるとき, 中心の y 座標の最小値]

xy 平面上の $y \geq x^2$ で表される領域を D とする.


D に含まれる 1 辺の長さ t の正方形で, 各辺が座標軸と平行または 45° の角をなすものを全て考える. このとき, これらの正方形の中心の y 座標の最小値を t の関数として表し, そのグラフをかけ.

(1983年理)



```
Clear[x, y, t]
Exists[x,
  t > 0 && ((y - t/2 ≥ (x - t/2)^2 && y - t/2 ≥ (x + t/2)^2) || (y - t/Sqrt[2] ≥ x^2 && y ≥ (x + t/Sqrt[2])^2 && y ≥ (x - t/Sqrt[2])^2)) ]
reg = Resolve[%, Reals]
RegionPlot[reg, {t, 0, 5}, {y, 0, 5}, PlotPoints → 50, Axes → True]
```

$$\exists_x \left(t > 0 \&\& \left(\left(-\frac{t}{2} + y \geq \left(x - \frac{t}{2} \right)^2 \&\& -\frac{t}{2} + y \geq \left(x + \frac{t}{2} \right)^2 \right) \vee \left(-\frac{t}{\sqrt{2}} + y \geq x^2 \&\& y \geq \left(x + \frac{t}{\sqrt{2}} \right)^2 \&\& y \geq \left(x - \frac{t}{\sqrt{2}} \right)^2 \right) \right) \right)$$

$$\left(t > 0 \&\& y \geq \frac{1}{4} (2t + t^2) \right) \vee \left(0 < t \leq \sqrt{2} \&\& y \geq \frac{t}{\sqrt{2}} \right) \vee \left(t > \sqrt{2} \&\& y \geq \frac{t^2}{2} \right)$$


[単位球内の2つの円盤が接している時, 半径の和の最小値]

xyz 空間において xy 平面上に円板 A があり


xz 平面上に円板 B があって

以下の 2 条件を満たしているものとする.

(a) A, B は原点からの距離が 1 以下の領域に含まれる.

(b) A, B は一点 P のみを共有し, P はそれぞれの円周上にある.

このような円板 A と B の半径の和の最大値を求めよ. ただし, 円板とは円の内部と円周をあわせたものを意味する. (1999 年理)



$P(p, 0)$, A, B の中心を $(a, b), (c, d)$. 半径を r_1, r_2 とおく. $(p-c)(p-b) \leq 0$ が「一点のみ」共有の条件.

```
Clear[a, b, c, d, r1, r2, p]
```

```
Exists[{a, b, c, d, r1, r2, p}, a^2 + (p - b)^2 == r1^2 && (p - c)^2 + d^2 == r2^2 && (p - c) (p - b) ≤ 0 && a^2 + b^2 ≤ (1 - r1)^2 && c^2 + d^2 ≤ (1 - r2)^2 && 0 < r1 ≤ 1 && 0 < r2 ≤ 1 && -1 < p < 1 && k == r1 + r2];
```

```
Resolve[%, Reals]
```

```
0 < k ≤ 9/8
```

§3. 領域

領域を表す式が求まらない, 或いは複雑すぎる時でも ContourPlot(3D), RegionPlot(3D) などを使って領域は図示できる.
また, 面積・体積も RegionMeasure を使って 求められる事が多い.

[折れ線の長さで距離を定義]


xy 平面上の2点P,Qに対し, PとQを x 軸または y 軸に平行な線分からなる折れ線で結ぶときの経路の長さの最小値を $d(P,Q)$ で表す.

(1) 原点(0,0) と点A(1,1) に対し, $d(O,P)=d(P,A)$ をみたす点P(x,y)


の範囲を xy 平面上に図示せよ.

(2) 実数 $a \geq 0$ に対し, 点 $Q(a, a^2 + 1)$ を考える. 次の条件(*)を満足する点P(x,y) の範囲を xy 平面上に図示せよ.

(*) $d(O,P)=d(P,Q)$ となるような $a \geq 0$ が存在する. (1994年理)




```
ClearAll[x1, x2, y1, y2, d, a, x, y]
d[{x1_, y1_}, {x2_, y2_}] = Abs[x1 - x2] + Abs[y1 - y2];
(*1*)
FullSimplify[d[{0, 0}, {x, y}] == d[{x, y}, {1, 1}]];
RegionPlot[%, {x, -2, 3}, {y, -2, 3}, Axes -> True]
```



```
(*2*)
Exists[a, a >= 0 && d[{0, 0}, {x, y}] == d[{x, y}, {a, a^2 + 1}]];
Resolve[%, Reals] // FullSimplify
RegionPlot[%, {x, -1, 1}, {y, 0, 2}]
```

$$(2y \geq 1 \ \&\& \ (x < 0 \ || \ (2x \geq 1 \ \&\& \ 1 + x^2 \geq x + 2y))) \ ||$$

$$(x \geq 0 \ \&\& \ ((1 + (2x + y)^2 > y \ \&\& \ 1 + x^2 < x + 2y) \ || \ (1 + x^2 \leq x + 2y \ \&\& \ 2y \leq 1))) \ || \ (2x \geq 1 \ \&\& \ 8y \geq 3 \ \&\& \ 2y \leq 1)$$


[参考]単なる式の簡略化の問題. 複雑な式でも軌跡の表示は簡単.


xy 平面上に次の2つの集合 l, m を考える.

$$l = \{(-5, y) \mid -5 < y < 5\} \quad m = \{(5, y) \mid -5 < y < 5\}$$

l, m 上にない2点 A, B に対し, A, B を l, m と交わらない線分または折れ線で結ぶときの経路の長さの最小値を $d(A, B)$ で表す. 2点 $P(-9, -3), Q(9, 3)$ に対し,


$$d(P, R) = d(Q, R)$$

となる点 R の軌跡を xy 平面上に図示せよ. (1993年理)



```
ClearAll["Global`*"];
a = {-5, -5}; b = {5, -5}; c = {5, 5}; d = {-5, 5}; p = {-9, -3}; q = {9, 3};
dP[x_, y_] = Which[(y ≥ 2 x + 15 || 2 y + x ≤ -15 || x ≤ -5), ArcLength[Line[{p, {x, y}}]], y ≥ 5 || y ≥ 1/2 Sqrt[x^2 + 10 x + 45],
  ArcLength[Line[{p, d, {x, y}}]], y < -5 || y < 1/2 Sqrt[x^2 + 10 x + 45] && x ≤ 5, ArcLength[Line[{p, a, {x, y}}]],
  y ≥ 1/2 Sqrt[x^2 - 10 x + 45], ArcLength[Line[{p, d, c, {x, y}}]], y < 1/2 Sqrt[x^2 - 10 x + 45], ArcLength[Line[{p, a, b, {x, y}}]]];
dQ[x_, y_] = dP[x, y] /. {x → -x, y → -y, p → q, q → p, a → c, c → a, b → d, d → b};
path = ContourPlot[dP[x, y] == dQ[x, y], {x, -12, 12}, {y, -12, 12}];

Manipulate[
  a = {-5, -5}; b = {5, -5}; c = {5, 5}; d = {-5, 5}; p = {-9, -3}; q = {9, 3}; x = r[[1]]; y = r[[2]];
  pts1 = Which[(y ≥ 2 x + 15 || 2 y + x ≤ -15 || x ≤ -5), {p, r}, y ≥ 5 || y ≥ 1/2 Sqrt[x^2 + 10 x + 45], {p, d, r},
    y < -5 || y < 1/2 Sqrt[x^2 + 10 x + 45] && x ≤ 5, {p, a, r}, y ≥ 1/2 Sqrt[x^2 - 10 x + 45], {p, d, c, r}, y < 1/2 Sqrt[x^2 - 10 x + 45], {p, a, b, r}];
  pts2 = Which[-y ≥ 15 - 2 x || -x - 2 y ≤ -15 || -x ≤ -5, {q, r}, -y ≥ 5 || -y ≥ 1/2 Sqrt[(-x)^2 + 10 (-x) + 45], {q, b, r}, -y < -5 ||
    (-y < 1/2 Sqrt[(-x)^2 + 10 (-x) + 45] && -x ≤ 5), {q, c, r}, -y ≥ 1/2 Sqrt[(-x)^2 - 10 (-x) + 45], {q, b, a, r}, -y < 1/2 Sqrt[(-x)^2 - 10 (-x) + 45], {q, c, d, r}];
  delta = ArcLength[Line[pts1]] - ArcLength[Line[pts2]];
  Show[
    Graphics[{Point[r], Thick, Line[{a, d}], Line[{b, c}], Thin, Blue, Line[pts1], Red, Line[pts2], Black,
      Point[p], Point[q], Text["P", p, {2, Center}], Text["Q", q, {-2, Center}],
      Text["R", r, {0, 2}], Text["A", a, {Center, Top}], Text["B", b, {Center, Top}], Text["C", c, {Center, Bottom}],
      Text["D", d, {Center, Bottom}], Text[Style["青線長-赤線長"] ==> ToString[Floor[100 N[delta]]/100., 9, Bold], {7, 12}],
    PlotRange → {{-13, 13}, {-13, 13}}, Axes → True, ImageSize → 240], path
  ], (*insert path*) {{r, {0, 0}}, Locator}]
```



[点の通過領域]


座標平面上の2点P, Qが、曲線 $y=x^2$ ($-1 \leq x \leq 1$) 上を自由に動くとき、
線分PQを1:2に内分する点Rが動く範囲を D とする。

ただし $P=Q$ のときは $R=P$ とする。

(1) a を $0 \leq a \leq 1$ を満たす実数とすると、点 (a, b) が

D に属するための b の条件を a を用いて表せ。

(2) D を図示せよ (2007年理)



1Tテキストに載っていた問題. Rが中点以外のときは複雑だが, QEなら簡単.

```
Clear[p, q, a, b]
```


```
Exists[{p, q}, -1 ≤ p ≤ 1 && -1 ≤ q ≤ 1 && 2 p + q == 3 a && 2 p^2 + q^2 == 3 b];
```

```
Resolve[%, Reals];
```

```
reg = FullSimplify[%]
```

```
RegionPlot[reg, {a, -1, 1}, {b, 0, 1}, PlotPoints → 50, Axes → True]
```

```
a ≥ -1 && a ≤ 1 && a^2 ≤ b && ((1 + 3 a^2 ≥ 2 (a + b) && 2 + a (4 + 3 a) ≥ b) || (2 + a (-4 + 3 a) ≥ b && 1 + a (2 + 3 a) ≥ 2 b))
```




[線分の通過領域]

点Oを原点とする座標空間内で、一辺の長さが1の正三角形OPQを動かす。また、点A(1,0,0)に対して、 $\angle AOP$ を θ とおく。ただし $0^\circ \leq \theta < 360^\circ$ とする。

(1) 点Qが(0,0,1)にあるとき、点Pのx座標がとりうる値の範囲と、 θ がとりうる値の範囲を求めよ。

(2) 点Qが平面 $x=0$ 上を動くとき、辺OPが通過しうる範囲を K とする。 K の体積を求めよ。(2017年理)



Pの軌跡(曲面)を求めてから線分OPの通過領域を求める。さもないとReduceでは一晩以上かかる。また式は途中でできるだけ簡単にした方がよい。「 $s^2 + t^2 == 1 \ \&\& \ q^2 + r^2 + p^2 == 1$ 」のとき「 $p^2 + (q - s)^2 + (r - t)^2 = 1 \Leftrightarrow 2qs + 2rt = 1$ 」このほうが実行時間が1/10になる。

```
Clear[p, q, r, s, t, x, y, z, u]
```

```
Exists[{s, t}, 2 q s + 2 r t == 1 && s^2 + t^2 == 1 && q^2 + r^2 + p^2 == 1];
```

```
surface = Resolve[%, Reals]
```

```
(q == -1/2 && q^2 + r^2 == 0 && p^2 + q^2 + r^2 == 1) || (q == 0 && r != 0 && r^2 >= 1/4 && p^2 + q^2 + r^2 == 1) ||
```

```
(q != 0 && r != 0 && q^2 + r^2 == 0 && p^2 + q^2 + r^2 == 1) || (q != 0 && q^2 + r^2 >= 1/4 && p^2 + q^2 + r^2 == 1) || (q == 1/2 && q^2 + r^2 == 0 && p^2 + q^2 + r^2 == 1)
```

```
Exists[{p, q, r, u}, x == u p && y == u q && z == u r && 0 < u < 1 && surface]; (*{p,q,r}と{p,q,r,u}の間違いに注意*)
```


```
Resolve[%, Reals]
```

```
body = ImplicitRegion[%, {x, y, z}];
```

```
(y == 0 && z != 0 && x^2 - 3 z^2 <= 0 && x^2 + z^2 < 1) || (y == 0 && z != 0 && x^2 - 3 z^2 == 0 && x^2 + z^2 < 1) ||
```

```
(y != 0 && z != 0 && y^2 + z^2 == 0 && x^2 + y^2 + z^2 < 1) || (y != 0 && x^2 - 3 y^2 - 3 z^2 <= 0 && x^2 + y^2 + z^2 < 1) || (y != 0 && x^2 - 3 y^2 - 3 z^2 == 0 && x^2 + y^2 + z^2 < 1)
```

```
RegionPlot3D[body, PlotPoints -> 50, PlotStyle -> Directive[Green, Opacity[0.5]], Mesh -> None]
```




RegionMeasure で体積は求まるが時間は掛かる (約4分)

```
RegionMeasure[body] // Timing
```

```
{123.912, 2 π / √3}
```

[点射影]

a, b, c を正の実数とする. xyz 空間において, $|x| \leq a$, $|y| \leq b$, $z = c$ を満たす点 (x, y, z) からなる板 R を考える. 点光源 P が平面 $z = c + 1$ 上の楕円 $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$, $z = c + 1$ の上を一周する時, 光が板 R にさえぎられて xy 平面上にできる影の通過する部分の図をえがき, その面積を求めよ. (1991年 東大理)



a, b, c のままでは図示できないので, $a = 2$, $b = 1$, $c = 3$ として解答する.

$P(X, Y, c + 1)$, R 上の点を $R'(x, y, c)$, 直線 PR' と xy 平面の交点を $Q(u, v, 0)$ とおくと, $\overrightarrow{QR'} = t \overrightarrow{QP} \Leftrightarrow \begin{pmatrix} x - u \\ y - v \\ c \end{pmatrix} = t \begin{pmatrix} X - u \\ Y - v \\ c + 1 \end{pmatrix}$


```
Clear[a, b, c, u, v, x, y, X, Y];
```

```
a = 2; b = 1; c = 3;
```

```
Exists[{x, y, X, Y, t}, Abs[x] ≤ a && Abs[y] ≤ b && {x - u, y - v, c} == t {X - u, Y - v, c + 1} && X^2 / a^2 + Y^2 / b^2 == 1];
```

```
pred = Resolve[%, Reals] // FullSimplify;
```

```
RegionPlot[pred, {u, -15, 15}, {v, -8, 8}, Axes → True, AspectRatio → Automatic]
```



面積は, 文字のままで求める

```
Clear[a, b, c, u, v, x, y, t, X, Y];
```

```
Exists[{x, y, X, Y, t},
```

```
a > 0 && b > 0 && c > 0 && Abs[x] ≤ a && Abs[y] ≤ b && {x - u, y - v, c} == t {X - u, Y - v, c + 1} && X^2 / a^2 + Y^2 / b^2 == 1];
```

```
pred = Resolve[%, Reals];
```

```
{time, area} = RegionMeasure[ImplicitRegion[pred, {u, v}]] // Timing // FullSimplify
```

```
{12.5581, {a b (4 + c (16 + c (12 + π))) c > 0 && a > 0 && b > 0
0 True}}
```

(おまけ)これらも 3行で 十分

1. 【楕円が円に含まれる条件】

α, β を正の数とし, xy 平面において,

$$\text{楕円 } C: \frac{x^2}{\alpha} + \frac{(y - \sqrt{\beta})^2}{\beta} = 1 \text{ と領域 } D = \{(x, y) \mid x^2 + y^2 \leq 1\}$$


を考える.

(1) C が D に含まれるような点 (α, β) の範囲

を求め, $\alpha\beta$ 平面上に図示せよ.

(2) 点 (α, β) が (1) で求めた範囲を動くとき,

楕円 C の面積の最大値を求めよ. (1996年 東大理)




```
(*1*) Clear[a, b]
```

```
ForAll[{x, y}, x^2/a + (y - Sqrt[b])^2/b == 1 && a > 0 && b > 0 => x^2 + y^2 <= 1];
```

```
reg = Resolve[%, Reals]
```

```
RegionPlot[reg, {a, 0, 1}, {b, 0, 1/2}]
```

$$a \leq 0 \mid \left(0 < a \leq \frac{1}{2} \text{ \&\& } b \leq \frac{1}{4} \right) \mid \left(\frac{1}{2} < a \leq 1 \text{ \&\& } b \leq a - a^2 \right) \mid (a > 1 \text{ \&\& } b \leq 0)$$



```
(*2*) Exists[{a, b}, Pi Sqrt[a b] == k && reg && a > 0 && b > 0];
```

```
Resolve[%, Reals] // FullSimplify
```

$$0 < k \leq \frac{2\pi}{3\sqrt{3}}$$

2. 【三角形が正方形に含まれる条件】

a, b を正の数とし, xy 平面上の2点 $A(a, 0)$ および $B(0, b)$ を頂点とする正三角形を ABC とする. ただし, C は第1象限の点とする.


(1) 三角形 ABC が正方形 $D = \{(x, y) \mid 0 \leq x \leq 1, 0 \leq y \leq 1\}$ に含まれるような (a, b) の範囲を求めよ.

(2) (a, b) が (1) の範囲を動くとき, 三角形 ABC の面積 S が最大となるような (a, b) を求めよ. また, そのときの S の値を求めよ. (1997年 東大理)

手で解くときは正三角形は回転で考える事が多いが, QEの場合は長さで考える.

```
dis[x_] := Total[x * x]; (*0とxの距離の2乗*)
```

```
(*1*)
Exists[{c1, c2}, dis[{a, 0} - {c1, c2}] == dis[{0, b} - {c1, c2}] == dis[{a, 0} - {0, b}] && 0 < a ≤ 1 && 0 < b ≤ 1 && 0 ≤ c1 ≤ 1 && 0 ≤ c2 ≤ 1];
reg = Resolve[%, Reals] // FullSimplify
RegionPlot[%, {a, 0, 1}, {b, 0, 1}]
(*2*) Exists[{a, b}, Sqrt[3] / 4 (a^2 + b^2) == k && reg];
Resolve[%, Reals]
```

$$0 < a \leq 1 \ \&\& \ 0 < b \leq 1 \ \&\& \ (-2 + a)^2 \geq 3 b^2 \ \&\& \ 3 a^2 \leq (-2 + b)^2$$


$$0 < k \leq -3 + 2\sqrt{3}$$

3. 【3 角関数を含んだ最小値】

実数 a, b に対して

$$f(\theta) = \cos 3\theta + a \cos 2\theta + b \cos \theta$$

とし, $0 < \theta < \pi$ で定義された関数

$$g(\theta) = \frac{f(\theta) - f(0)}{\cos(\theta) - 1}$$

を考える.

(1) $f(\theta)$ と $g(\theta)$ を $x = \cos \theta$ の式で表せ.

(2) $g(\theta)$ が $0 < \theta < \pi$ の範囲で最小値 0 をとるための a, b についての条件を求めよ.
また, 条件を満たす点 (a, b) が描く図形を座標平面上に図示せよ. (2017年理)


三角関数も置き換えて多項式になる場合がある. そのときはQEで解ける.

```
(*1*) Clear[a, b, x]
TrigExpand[Cos[3 θ] + a Cos[2 θ] + b Cos[θ]]
f[x_] = % /. {Sin[θ] → Sqrt[1 - x^2], Cos[θ] → x} // Expand;
g[x_] = Simplify[(f[x] - f[1]) / (x - 1)];
Print["f(θ)=", TraditionalForm[f[x]], " ", "g(θ)=", TraditionalForm[g[x]]]

b Cos[θ] + a Cos[θ]^2 + Cos[θ]^3 - a Sin[θ]^2 - 3 Cos[θ] Sin[θ]^2
f(θ) = 2 a x^2 - a + b x + 4 x^3 - 3 x      g(θ) = 2 a (x + 1) + b + (2 x + 1)^2

(*2*) p = ForAll[x, (0 < x < 1 ⇒ g[x] ≥ 0)] && Exists[y, g[y] == 0 && 0 < y < 1];
(*p=Exists[y,g[y]==0&&0<y<1&&ForAll[x,0<x<1⇒g[x]≥0]]もOK*)
r = Resolve[p, Reals] // FullSimplify
RegionPlot[ImplicitRegion[r, {a, b}]]
(*ContourPlot[a^2==4 (a+b), {a,-20,20},{b,-20,20},
  RegionFunction→Function[{a,b},((1+2 a+b>0&&-6<a≤-4)|| (9+4 a+b>0&&-4<a<-2)))]*)
(*↑[別法] ContourPlotを使い RegionFunctionで (a,b)の範囲を制限する*)
```

$$a^2 = 4(a+b) \quad \&\& \quad -6 < a < -2 \quad \&\& \quad 9 + 4a + b > 0$$



[積分を含んだ最大値]

$0 \leq \alpha \leq \beta$ をみたす実数 α, β と, 2 次式 $f(x) = x^2 - (\alpha + \beta)x + \alpha\beta$ について $\int_{-1}^1 f(x)dx = 1$ が成立しているとする. このとき定積分 $S = \int_0^\alpha f(x)dx$ を α の式で表し, S がとりえる値の最大値を求めよ. (2008年文)

4.

積分計算してしまえば多項式の問題なのでQEで解ける.

```
Clear[x]
```

```
Exists[{α, β}, 0 ≤ α ≤ β && Integrate[(x - α) (x - β), {x, -1, 1}] == 1 && Integrate[(x - α) (x - β), {x, 0, α}] == S];
```

```
Resolve[%, Reals]
```

$$0 < S \leq \frac{1}{18\sqrt{6}}$$

[虚部が負となる条件]

x についての方程式

$$px^2 + (p^2 - q)x - (2p - q - 1) = 0$$

が解を持ち, すべての解の実部が負となるような

実数の組 (p, q) の範囲を pq 平面上に図示せよ. (1992年文)

5.

虚数を扱うときは `Resolve[Complexes]` と `a ∈ Reals` を組み合わせる.

```
Clear[x1, x2, p, q]
```

```
Exists[{x1, x2}, (p | q) ∈ Reals && x1 + x2 == -(p^2 - q) / p && x1 * x2 == -(2 p - q - 1) / p && p ≠ 0 && Re[x1] < 0 && Re[x2] < 0];
```

```
A = Resolve[%, Complexes] // FullSimplify;
```


```
Exists[x, -q x + q + 1 == 0 && p == 0 && x < 0];
```

```
B = Resolve[%, Reals] // FullSimplify;
```

```
FullSimplify[A || B]
```

```
RegionPlot[A || B, {p, -2, 2}, {q, -2, 4}, PlotPoints → 50, Axes → True]
```

$$(2p < 1 + q \&\& p > 0 \&\& p^2 > q) \mid \mid (p = 0 \&\& -1 < q < 0)$$




6. 【パラメータの多い通過領域】

平面上に2定点A,Bがあり，線分ABの長さ $\overline{AB}$ は $2(\sqrt{3}+1)$ である．

この平面上を動く3点P,Q,Rがあつて，つねに $\overline{AP}=\overline{PQ}=2$ ，

$\overline{QR}=\overline{RB}=\sqrt{2}$ なる長さを保ちながら動いている．このとき点Qが動きうる範囲を図示し，その面積を求めよ．(1982年東大文)



残念ながら領域を表す式は綺麗にならないが，図示はでき，その上，面積も求まる．

```
Clear[p1, p2, r1, r2, x, y]
```

```
a = 2 (Sqrt[3] + 1);
```


```
Exists[{p1, p2}, p1^2 + p2^2 == 4 && (p1 - x)^2 + (p2 - y)^2 == 4] && Exists[{r1, r2}, (r1 - a)^2 + r2^2 == 2 && (x - r1)^2 + (y - r2)^2 == 2];
```

```
Resolve[%, Reals];
```

```
reg = FullSimplify[%]
```

$$\left(y = 0 \&\& \left(x = 4 \mid 2\sqrt{2} + x = 2 + 2\sqrt{3} \right) \right) \mid \mid \left(x > 0 \&\& \left(\left(2\sqrt{2} + x > 2 + 2\sqrt{3} \&\& \right. \right. \right. \\ \left. \left. \left. \text{Root} \left[320 + 192\sqrt{3} - 288x - 160\sqrt{3}x + 88x^2 + 48\sqrt{3}x^2 - 8x^3 - 8\sqrt{3}x^3 + x^4 + \left(24 + 16\sqrt{3} - 8x - 8\sqrt{3}x + 2x^2 \right) \mp 1^2 + \mp 1^4 \&, 1 \right] \leq y \leq \right. \right. \right. \\ \left. \left. \left. \text{Root} \left[320 + 192\sqrt{3} - 288x - 160\sqrt{3}x + 88x^2 + 48\sqrt{3}x^2 - 8x^3 - 8\sqrt{3}x^3 + x^4 + \left(24 + 16\sqrt{3} - 8x - 8\sqrt{3}x + 2x^2 \right) \mp 1^2 + \mp 1^4 \&, 2 \right] \&\& \right. \right. \right. \\ \left. \left. \left. \left(2 + \sqrt{3} \right) x < 6 + 4\sqrt{3} \right) \mid \mid \left(\left(2 + \sqrt{3} \right) x \geq 6 + 4\sqrt{3} \&\& \sqrt{16 - x^2} + y \geq 0 \&\& x < 4 \&\& y \leq \sqrt{16 - x^2} \right) \right) \right)$$

```
RegionPlot[reg, {x, 2, 5}, {y, -2, 2}, Axes -> True]
```



```
RegionMeasure[ImplicitRegion[reg, {x, y}]] // FullSimplify
```

$$-4 \left(1 + \sqrt{3} \right) + \frac{14\pi}{3}$$

7. [立体射影]

xyz 空間内の原点を中心とする半径 1 の球面


$$S = \{(x, y, z) \mid x^2 + y^2 + z^2 = 1, x, y, z \text{ は実数}\}$$

を考え, S 上の定点 $(0, 0, 1)$ を A とする.

A とことなる S 上の点 $P(x, y, z)$ に対し,
直線 AP と xy 平面の交点を $Q(u, v, 0)$ とする.

$$\text{点 } P \text{ が } x^2 + y^2 + z^2 = 1, x \geq \frac{1}{3}, y \geq \frac{1}{3}, z \geq \frac{1}{3}$$

を満たしながら動くとき, 対応する点 Q の動く
範囲を uv 平面上に図示せよ. (1993年 東大文 改)



```
Clear[k, u, v, x, y, z]; k = 3; (*オリジナルは k のまま*)
Exists[{x, y, z, t}, x^2 + y^2 + z^2 == 1 && x >= 1/k && y >= 1/k && z >= 1/k && {x, y, z - 1} == t {u, v, -1} && 0 < t < 1];
pred = Reduce[%, {u, v}, Reals] // FullSimplify
(2 u == 1 && 2 v == Sqrt[7]) || (2 u == 3 + Sqrt[7] && 2 v == 3 + Sqrt[7]) ||
(Sqrt[-1 - (-6 + u) u] >= v && ((2 u > 1 && 2 u <= Sqrt[7] && Sqrt[2 - u^2] <= v) || (2 u > Sqrt[7] && Sqrt[8 - u^2] + v >= 3 && 2 u < 3 + Sqrt[7])))

reg = ImplicitRegion[pred, {u, v}];
RegionPlot[reg]
```

